

Appendix C. Field methods and analyses of the butterfly surveys of the Aargau Biodiversity Monitoring Program.

FIELD METHODS

In the canton of Aargau in northern Switzerland, butterflies were surveyed on 519 study plots (the same plots were used in the bird surveys, see Appendix B). At each study plot, the butterflies were surveyed using standardized transect counts (Pollard et al. 1995); the length of the transects was 250 m, and butterflies were recorded within 5 meters on one side of the transect line on the way forth and on the other side on the way back. Each examined transect was visited 11 times between April 21 and September 21, thus the entire flight periods of most butterfly species were covered. For a more detailed description of the butterfly surveys and the collection of land-use information for the transects, see Altermatt (2012).

MODEL DESCRIPTION

From 1998 to 2010 totally 519 study plots were sampled every five years during 11 visits for the presence of a butterfly species, resulting in 1'337 observed occupancy histories. For instance, an observed occupancy history 0,0,0,1,1,0,0,0,0,0,0 indicates that on the respective plot and during the respective year, the butterfly species was observed only during the fourth and fifth of totally 11 visits. The data matrix $y[o,j]$ contains the $o=1,...,1'337$ observed occupancy histories of the $j=1,...,11$ visits, the vector $year[o]$ contains the year when the observed occupancy history o was sampled, vector $site[o]$ contains the id of the study plot, the vector $land_use[o]$ contains the land-use type (1=residential area, 2=forests, 3=agriculture, 4=mixed land-use) of study plot $site[o]$, and the matrix $DATE[o,j]$ contains the date (i.e. the number of days after 31 March) when the j th visit was conducted to study plot $site[o]$.

We define the true occupancy state $x[o]$ of the observed occupancy histories $o=1,...,1'337$ such that $x[o]=0$ if the plot of the observed occupancy history o is never occupied during the entire season, and $x[o]=1$ otherwise. We assume that the true occupancy state $x[o]$ is a Bernoulli random variable with site-occupancy ψ_o as its parameter

$$x[o] \sim \text{Bernoulli}(\psi_o)$$

The study plots are assumed to become occupied at different dates within the season. This arrival process is described with an overdispersed Poisson distribution with average arrival date ϕ and a standard deviation σ . If $z[o]$ is defined as the first arrival date of a butterfly species then

$$z[o] \sim \text{Poisson}(\mu_o) \text{ with } \mu_o \sim \text{Normal}(\phi_o, \sigma)$$

Further, we assume that the species depart from the study plots at different dates within the season. Like the arrival process, we describe the departure process with an overdispersed Poisson distribution with average departure date λ and a standard deviation ξ . If $d[o]$ is defined as the departure date of a species then

$$d[o] \sim \text{Poisson}(\omega_o) \text{ with } \omega_o \sim \text{Normal}(\lambda_o, \xi)$$

If we define p as the probability that the species present at a study plot is observed, the observed occupancy history $y[o,j]$ (i.e. the data) is given as

$$y[o,j] \sim \text{Bernoulli}(x[o] * p_o * I[o,j])$$

$$\text{with } I[o,j] = \begin{cases} 1 & \text{when } z[o] \leq \text{DATE}[o,j] \text{ and } d[o] > \text{DATE}[o,j] \\ 0 & \text{when } z[o] > \text{DATE}[o,j] \text{ or } d[o] \leq \text{DATE}[o,j] \end{cases}$$

Thus, the entire model is described by the four parameters that are specific for an occupancy history, site-occupancy ψ_o , detection probability p_o , average arrival ϕ_o and average departure λ_o , by the standard deviation σ of arrival dates of species at the different study plots, and by the standard deviation ξ of departure dates of species at the different study plots. We then modeled the four parameters as

$$\text{logit}(\psi_o) = \alpha_{\text{site}[o]} + a_{1,\text{land_use}[o]} + a_{2,\text{land_use}[o]} \text{year}[o]$$

where $\alpha_{site[o]}$ is the random site effect, $a_{1,land_use[o]}$ the land-use type specific intercept and $a_{1,land_use[o]}$ the land-use type specific temporal trend of the site-occupancy ψ_o ,

$$\text{logit}(p_o) = b_{1,j} + b_{2,j} \text{year}[o]$$

where $b_{1,j}$ is the intercept specific for a visit and $b_{2,j}$ the temporal trend of the detection probability p_o specific for a visit, and

$$\phi_o = c_{1,j} + b_{2,j} \text{year}[o]$$

where $c_{1,land_use[o]}$ is the intercept specific for a land-use type and $c_{1,land_use[o]}$ the temporal trend of the average arrival ϕ_o specific for a land-use type,

$$\lambda_o = d_{1,j} + d_{2,j} \text{year}[o]$$

where $d_{1,land_use[o]}$ is the intercept specific for a land-use type and $d_{1,land_use[o]}$ the temporal trend of the average departure λ_o specific for a land-use type.

In Fig. C1, the entire model is given in the BUGS language (Link et al. 2002).

```

model {
### Define priors
# Site-occupancy
for(u in 1:4) {
  a[1,u] ~ dnorm(0, 0.01)
  a[2,u] ~ dnorm(0, 0.01)
}
sigma.alpha ~ dgamma(1,1)
tau.alpha <- 1/(sigma.alpha * sigma.alpha)
for(i in 1:nsites) {
  alpha[i] ~ dnorm(0, tau.alpha)
}
for(o in 1:nobs) {
  mu.psi[o] <- a[1,land_use[o]]+a[2,land_use[o]]*year[o]+alpha[site[o]]
  mul.psi[o] <- min(10, max(-10, mu.psi[o]))
  logit(psi[o]) <- mul.psi[o]
}
# Detection probability
mu.b1 ~ dnorm(0, 0.01)
sigma.b1 ~ dgamma(1,1)
tau.b1 <- 1/(sigma.b1 * sigma.b1)
mu.b2 ~ dnorm(0, 0.01)
sigma.b2 ~ dgamma(1,1)
tau.b2 <- 1/(sigma.b2 * sigma.b2)
for(j in 1:J) {
  b[1,j] ~ dnorm(mu.b1, tau.b1)
  b[2,j] ~ dnorm(mu.b2, tau.b2)
}
for(t in 1:nyears) {
  for(j in 1:J) {
    tp[t,j] <- b[1,j] + b[2,j]*(t-1)
    lp[t,j] <- max(-10, min(10, tp[t,j]))
    logit(p[t,j]) <- lp[t,j]
  }
}
# Arrival date
mu.c1 ~ dnorm(90, 0.001)
sigma.c1 ~ dgamma(1,1)
tau.c1 <- 1/(sigma.c1 * sigma.c1)
mu.c2 ~ dnorm(0, 0.01)
sigma.c2 ~ dgamma(1,1)
tau.c2 <- 1/(sigma.c2 * sigma.c2)
sigma.over ~ dgamma(1,1)
tau.over <- 1/(sigma.over * sigma.over)
for(u in 1:4) {
  c[1,u] ~ dnorm(mu.c1, tau.c1)
  c[2,u] ~ dnorm(mu.c2, tau.c2)
}
for(o in 1:nobs) {
  eps[o] ~ dnorm(0, tau.over)
  mu.phi[o] <- c[1,land_use[o]] + c[2,land_use[o]]*year[o] + eps[o]
  mul.phi[o] <- min(200, max(30, mu.phi[o]))
  z[o] ~ dpois(mul.phi[o])
}
# Departure date
mu.d1 ~ dnorm(120, 0.001)
xi ~ dgamma(1,1)
tau.d1 <- 1/(sigma.d1 * sigma.d1)
mu.d2 ~ dnorm(0, 0.01)
sigma.d2 ~ dgamma(1,1)
tau.d2 <- 1/(sigma.d2 * sigma.d2)
sigma.d.over ~ dgamma(1,1)
tau.d.over <- 1/(sigma.d.over * sigma.d.over)
for(u in 1:4) {
  d[1,u] ~ dnorm(mu.d1, tau.d1)
  d[2,u] ~ dnorm(mu.d2, tau.d2)
}
for(o in 1:nobs) {
  d.eps[o] ~ dnorm(0, xi)
  mu.lambda[o] <- d[1,land_use[o]] + d[2,land_use[o]]*year[o] + d.eps[o]
  mul.lambda[o] <- min(500, max(0, mu.lambda[o]))
  d[o] ~ dpois(mul.lambda[o])
}
### Arrival and departure model
for(o in 1:nobs) {
  x[o] ~ dbern(psi[o])
  for(j in 1:J) {
    mu[o,j] <- x[o]*p[year_index[o],j]*step(DATE[o,j]-z[o])*step(d[o]-DATE[o,j])
    mul[o,j] <- min(0.99, max(0.01, mu[o,j]))
    y[o,j] ~ dbern(mul[o,j])
  }
}
}

```

Fig. C1. The model in BUGS-language

BAYESIAN ANALYSES

For Bayesian analysis, we used MCMC methods conducted using OpenBugs 3.2.1 (Thomas et al. 2006) and executed in R using the R add-on library R2OpenBUGS (Sturtz et al. 2005). We used vague priors for all parameters (see Fig. C1 for exact specification of the priors), and posteriors were based on two parallel chains with 30'000 iterations each, discarding the first 20'000 values and thinning the remainder by using every 10th value. We assessed convergence using the Gelman–Rubin diagnostic (Brooks and Gelman 1998). We used the means of the simulated values of the posterior distributions as point estimates of the parameters and the 2.5% and 97.5% quantiles as estimates of the credible intervals. We speak of a "clear" effect (which, in a frequentist terminology, may be similar to a significant effect) if zero was not included in the 95% Bayesian credible interval of an estimate, or of a "clear" difference between two land-use categories if the point estimate for the first land-use category was not within the credible-interval of the estimate for the second land-use category (Amrhein et al. 2012).

RESULTS ON SITY-OCCUPANCY AND DETECTION PROBABILITY

In the canton of Aargau, the Ringlet (*Aphantopus hyperantus*) mainly occurred in forest, agriculture and mixed land-use plots, while the Marbled White (*Melanargia galathea*) almost exclusively occurred in agricultural plots (Fig. C2a). Between 1998 and 2010, there was a clear positive trend in site-occupancy of Ringlets in forest habitats, while the species tended to decrease in the other habitats. In agricultural land, where most Marbled Whites were detected, the number of occupied plots clearly increased over the study period (Fig. C2b). For both species, there was considerable variation in detection probability over the season and also in the temporal trend over the study period (Fig. C3). The results on the estimated arrival times are given in the main body of the manuscript.




Fig. C2. Site-occupancies in 2005 (a), and temporal trends (1998-2010) of site-occupancies (b) for the summer flying butterflies Ringlet (*Aphantopus hyperantus*) and Marbled White (*Melanargia galathea*) in residential areas (●), forests (▲), agriculture (■) and mixed land-use plots (◆) as estimated from the model. Given are means and 95% credible intervals of the posterior distributions.




Fig. C3. Detection probabilities in 2005 (a), and temporal trends (1998-2010) of detection probabilities (b) for the summer flying butterflies Ringlet (*Aphantopus hyperantus*) and Marbled White (*Melanargia galathea*) during the 11 visits to the study plots. Given are means and 95% credible intervals of the posterior distributions.

LITERATURE CITED

- Altermatt, F. 2012. Temperature-related shifts in butterfly phenology depend on the habitat. *Global Change Biology* 18:2429-2438.
- Amrhein, V., B. Scaar, M. Baumann, N. Minery, J. P. Binnert, and F. Korner-Nievergelt. 2012. Estimating adult sex ratios from bird mist netting data. *Methods in Ecology and Evolution* 3:713-720.
- Brooks, S. P., and A. Gelman. 1998. General methods for monitoring convergence of iterative simulations. *Journal of Computational and Graphical Statistics* 7:434-455.
- Link, W. A., E. Cam, J. D. Nichols, and E. G. Cooch. 2002. Of BUGS and birds: Markov chain Monte Carlo for hierarchical modeling in wildlife research. *Journal of Wildlife Management* 66:277-291.
- Pollard, E., D. Moss, and T. J. Yates. 1995. Population trends of common british butterflies at monitored sites. *Journal of Applied Ecology* 32:9-16.
- Sturtz, S., U. Ligges, and A. Gelman. 2005. R2WinBUGS: A package for running WinBUGS from R. *Journal of statistical Software* 12:1-16.
- Thomas, A., R. B. O'Hara, U. Ligges, and S. Stutz. 2006. Making BUGS open. *R News* 6:12-17.